RHEL/R 134

Copy 1

RHEL/R 134
Rutherford Laboratory Report

Rutherford Laboratory DDP-224 System Handbook

D H Lord

Science Research Council

Track Analysis Group
Rutherford High Energy Laboratory
Chilton Didcot Berkshire
1966

Available from HMSO
price 11s net

Printed by Office Services (Luton) Ltd

RHEL/R 134

SCIENCE RESEARCH COUNCIL

RUTHERFORD LABORATORY DDP-224 STSTEM HANDBOOK

D H Lord*

ABSTRACT

A system of special purpose peripheral equipment has been built up around the double computer complex of a DDP-224 and an Orion This system is described in detail, and to aid those who wish to use it basic program examples are given of how to operate all parts of the system.

Track Analysis Group
Rutherford High Energy Laboratory
Chilton Didcot Berkshire

*Address with effect from June 1966
D.D. Division
CERN Geneva

April 1966

CONTENTS

Page No.

1.0 Introduction

In May 1965 a DDP-224 computer was delivered to the Rutherford Laboratory. Since the installation of this computer an "on-line" system involving a number of measuring machines and a sonic spark chamber experiment, has been built up around the Orion and DDP computers. The objectives of this system are:-

i) To provide a means of merging the data from a number of different "on-line" sources operating simultaneously and then forwarding it to the Orion.

ii) To provide the facility to do varying degrees of preliminary computation on this data.

iii) To provide storage to buffer the data flow into the Orion.

iv) To provide the means of controlling various "on-line" devices.

v) To provide typewriters and visual displays, to control programs, and to give information on the performance of the various on-line devices.

vi) To provide the means of writing on to, and reading from, IBM compatible magnetic tape.

vii) To provide the means of operating the "on-line devices" when the Orion computer is not available.

It is hoped that this manual will aid people wishing to write programs for this system, to add new devices to the system or to modify the existing system. The sections relating to the various devices can be read independently. This manual should be read in conjunction with the:

DDP-224 Programmers Manual

DAP II Manual

Main Frame Schematics Option Schematics Modification Schematics	For readers interested in hardware details

2.0 Configuration

The equipment delivered by the computer manufacturers was:

a)	DDP-224 computer with 12 K words of core store and three index registers.	1 off
b)	16 level priority interrupt system, with enable/disable of each interrupt; four levels are single execute interrupts.	1 off
c)	Interrupt Range Register.	1 off
d)	Power failure interrupt.	1 off
e)	Direct Memory Access Channel. (D.M.A.)	2 off
f)	Parallel unbuffered input channel.	2 off
g)	Parallel unbuffered output channel.	1 off
h)	Character buffer input/output channel.	2 off
i)	Output Control Pulses (OCP's).	24 off
j)	Sense Lines (SKS's).	16 off
k)	Console IBM typewriter.	1 off
l)	Remote IBM typewriter.	1 off
m)	Facit paper tape reader.	1 off
n)	Facit paper tape punch.	1 off
o)	Magnetic tape deck.	2 off
p)	Calcomp Graph Plotter.	1 off
q)	Digital C.R.T. Display.	1 off

In building up the on-line system the following equipment has been attached to the DDP-224 (See Fig. 1):

a) The Orion computer.

b) The Hough Powell Device (HPD) - a flying spot scanner of bubble chamber pictures.

c) A C.R.T. scanner of spark chamber film.

d) A sonic spark chamber experiment system (OLX).

e) **A storage CRT digital display, located in the control room of the spark chamber experiment.**

f) **A teletype teleprinter located in a control room in the Nimrod experimental area.**

The system is shown schematically in Figures 1 - 4, which correspond to Drawing Nos. CR 023667 to 023670 respectively. The remaining drawings relevant to this report are Nos. CR 023645 - 49, 52-54 and 59-66, copies of which may be obtained on request.

3.0 Core Store

The 12K words of the Rutherford Laboratory DDP-224 is divided into two units: an 8K module (modules A and B) and a 4K module (module C). Because 12K ($30,000_8$) is not a power of 2, this gives a rather curious "wrap-around" of addresses.

For example:

LDA '27777

would put the contents of location $27,777_8$ into A,

but

LDA '30000

would do $C(20,000_8) \rightarrow A$

and

LDA '40000

would do $C(00,000_8) \rightarrow A$

etc.

Details of the relationship between an address and the actual location in the core store that it refers to are given in Table 5.

Clearly care must be taken when using address modification with locations near the top of the core store.

Provided the computer is halted when power is switched off, the contents of the core store are preserved. If the Master Clear button is held down whilst power is switched on the contents of the core store are again unchanged.

4.0 Interrupts

The interrupts system provides the means whereby a peripheral device can force the computer to execute (at the completion of the current instruction) a jump to a specific location associated only with that interrupt. At that location there is normally a further jump instruction to a routine that will service the device, e.g. when the CRT scanner completes a scan line it generates an interrupt, which causes the DDP-224 to execute a routine which terminates the current data transfer from the CRT and starts a new one.



The action of an interrupt is in fact to force the computer to execute a Jump Store (JST) instruction to the "trap location" (even address) assigned to the interrupt. Thus if the last instruction to be executed, before the interrupt takes effect, was in location P, then (P+1) is stored in the trap location and next instruction to be executed is in the "transfer location" (odd address) of the interrupt. At the completion of the "interrupt routine" it is necessary to return to the original program at location P+1 and to cancel the interrupt condition, this is accomplished by:-

JRT TL - where TL is the Trap Location

The JRT is equivalent to a JMP* TL.

4.1 Priority Level System

There are sixteen interrupt lines fitted to the RHEL DDP-224. These are in a priority level system, i.e. each interrupt is assigned to one of sixteen priority levels, $L_{\emptyset\emptyset}$ to L_{17}. After receiving an interrupt on level N, and whilst executing the associated interrupt routine, the computer is in the interrupt level N condition. It cannot be interrupted by interrupts assigned to lower levels or another interrupt on level N, L_N ...... $L_{\emptyset\emptyset}$, but it can be interrupted by interrupts on higher levels, L_{17} .... L_{N+1}.

At the end of an interrupt routine, when the JRT instruction is executed the interrupt level N condition is changed to the interrupt level of the routines to which the JRT instruction is returning. An example of a situation with three active interrupts on levels L_2, L_6, L_{12} might be as follows:



1) A level 2 interrupt occurs causing a jump to the level 2 routine.

2) At the completion of the level 2 routine, control returns to the main program.

3) A level 2 interrupt causes a jump to level 2 routine.

4) Whilst at level 2 a level 6 interrupt occurs, causing a jump to the level 6 routine.

5) A level 12 interrupt causes a jump out of the level 6 routine to the level 12 interrupt routine.

6), 7), 8) At the completion of the interrupt routines control returns in sequence down to the main program.

9) A level 2 interrupt causes a jump directly to the level 2 routine.

10) A level 12 interrupt causes a jump directly to the level 12 routine.

11) Whilst in level 12 routine a level 6 interrupt occurs, it has to wait.

12) At the completion of the level 12 routine, control returns to the level 2 routine, but as there is a level 6 interrupt waiting, a jump occurs immediately to the level 6 routine.

13) As the level 6 and 2 routines are completed control returns in sequence down to the main program.

The assignment of interrupts to priority levels is shown in Table 3.

4.2 Hardware modifications to priority system

As the DDP-224 was originally supplied, it was necessary for a device to maintain its interrupt signal "on" until an acknowledge (e.g. ACK Ø6) signal, from the interrupt system, was received. This acknowledge signal could be delayed by an arbitrary length of time, in particular when an interrupt had to wait for higher level routines to be completed (see (12) above). To eliminate this necessity for the peripheral devices to deal with the acknowledge signal etc., a set of interrupt storage flip-flops were added (see CR 023652). This modification provides the following features.

i) It has twelve flip-flop, which can provide interrupts on levels $L_{ØØ}$ to L_{13}. These flip-flops are reset by the corresponding acknowledge signals.

ii) It provides a means of assigning the various interrupt signals from devices, to the required interrupt levels, by making suitable connections on a "jumper card" (3F2A29).

iii) The device needs only provided a transition of -6v to 0v, with a rise time of 200 nsec on its interrupt line.

iv) If an interrupt is disabled, then the corresponding "interrupt" storage flip-flop" is cleared and cannot be set.

4.3 Single Execute Interrupts

The four interrupts with highest priority (L_{17}, L_{16}, L_{15}, L_{14}) are single execute interrupts. The effect of a single execute interrupt is to cause the execution of just the instruction in the interrupt's trap location (unless it is a jump instruction), the program being executed when the interrupt took effect, then continues with its next instruction.

If an IRX instruction is executed by a single execute interrupt, then the index registers and the A register are not affected.

If an INM or OTM instruction is executed, the address field of the actual instruction is incremented by one each time it is executed (see 6.9.1.d).

4.4 Interrupt Enable/Disable System

Besides turning the entire interrupt system on and off by the ENB and INH instructions or by the console switch, it is possible to enable or disable each of the interrupts independently. This is accomplished by transferring a 16 bit mask to the interrupt system mask register via the Output Bus. The 16 bit mask can be transferred either from the A Register (bits 9 - 24) or from a core location. Bit 9 corresponds to interrupt L_{17}, bit 24 to $L_{\emptyset\emptyset}$. A suitable program sequence to enable, for example, interrupts L_1, L_5, L_{16}, would be:

```
        INH                 Inhibit interrupts
        OCP '50             Connect O/P Bus to interrupt mask register
        OTM MASK            Write mask into mask register
        ENB                 Enable interrupts
        - - - - - - - -
        - - - - - - - -
   MASK  PZE  '137735
```

Note

1. A "1" in a bit position in the mask disables the associated interrupt, "0" enables the interrupt.
2. The interrupt mask must not be changed whilst the system as a whole is enabled. Therefore the above sequence starts with an INH and finished with an ENB instruction.

4.4.1 Modifications to Enable/Disable System

The Enable/Disable system has been modified so that, like the parallel channels etc., the status of the mask transfer via the O/P Bus can be stored and controlled by the LDK/STK-System (See Section 5.1).

4.5 Interrupt when the CPU is Halted

Provided that the interrupt system has been suitably enabled, the DDP-224 will respond to interrupts even though it is in the Halt condition. When an interrupt is received the computer executes a "forced JST" as usual to the trap location and then executes the associated interrupt routine. After the JRT at the end of the interrupt routine the computer returns to its former Halted state. The priority interrupt system operates as normal, allowing interrupting of interrupt routines by interrupts of higher priority.

Note

The HLT instructions must only be in the routines at the no interrupt level, there must be no HLT's in interrupt routines.

5.0 I/O Channels

The channels of the DDP-224 can be divided into two broad categories.

A) Direct Memory Access (DMA) Channels. In these channels the data is transmitted directly to and from the core store via the Memory Interface (see Fig. 1).

B) All other channels, such as Parallel I/P, DMA set-up channel, Character Buffer, set-up channel for Interrupt Mask etc. All of these channels transmit data to and from the CPU (and core store) via the Input and Output Buses. As all these channels share essentially a single data transmission system, then only one channel can be enabled at a given time. Enabling a channel disables all other channels in this category (See Fig. 1).

The channels in category B each have essentially, two main control flip-flops:

Enable (EN) - These flip-flops determine which channel is currently selected to transmit data via the I/P or O/P-Bus.

Ready (RDY), or Not Busy ($\overline{\text{BSY}}$ or $\overline{\text{BUS}}$). - These flip-flops are used to synchronise the transfers of words between the C.P.U. and the peripheral devices and their control units or interfaces. When the RDY (or $\overline{\text{BSY}}$) is set, it indicates to the CPU that the channel is ready to transfer a word to (or from) the CPU. When it is reset, it indicates that the channel and device either haven't a word ready to transfer into the CPU or are still busy with the last word received from CPU and are not ready to receive a new word.

In the case of the 'Set-Up' channels of the DMA's, Interrupt Range Register, and the read DMA range register channels, these two control flip-flops are amalgamated into a single control flip-flop e.g. PKFEI.

The DMA's (category A) have essentially one control flip-flop (Ready-IORG) and a control clock (FTØ1 to FTØ4).

5.1 Channel Status (STK/LDK) System

The use of the channels in category B (see above) with interrupts produces difficulties. For example consider the situation where the computer is operating with a main programme that contains transfers from the paper tape reader, in addition there is a peripheral device active on a parallel input channel. The peripheral device produces interrupts which causes entry to an interrupt routine which arranges the transfers from the device through the channel. From time to time the interrupt will occur when a transfer involving the reader is in progress. The interrupt routine has to enable the parallel I/P channel, thus disabling the standard character buffer. It is therefore necessary for the interrupt routine to be able to determine whether the standard character buffer is enabled, before it enables another channel and at the end of the routine, re-enable the character buffer if necessary.

This is accomplished by the Store Channel Status (STK) and Load Channel Status (LDK) instructions, in conjunction with the A Register. The assignment of the channel enable flip flops etc. to bits in the A Register is shown in Table 4.

Note

1. The Read DMA Range Register enable flip-flops are not wired into this system as there was no spare circuitry available.
2. The TAPIN (magnetic tape selected for I/P) flip-flop is in the LDK/STK system because inward tape transfers uses the I/P Bus, so

that it is effectively an Enable Flip-flop.

The STK instruction sets the corresponding bits in the A-Register of all enabled channels. LDK enables all channels whose corresponding bits in the A-Register are set to "1".

Another valuable use of LDK is to enable channels without changing the state of their channel ready flip-flops. For example, the H.P.D. has a word set up on its output lines to Parallel I/P channel #1, it will have set the channel RDY flip-flop. If the I/P channel is now enabled by an OCP, in the usual way, the RDY flip-flop is reset, thus to the HPD the word would appear to have been read, although it had not. If LDK is used,.the state of the RDY flip-flop is unchanged, thus no words are lost.

6.0 Peripheral Devices

The operation of the various standard and special peripherals attached to the DDP-224 will now be described.

6.1 Console Typewriter

This is an IBM "Golf-Ball" typewriter, capable of up sec. It is attached to the Standard Character Buffer (S.C.B) via a Control Unit. The Control Unit takes care of all the timing and coding changes needed to operate the typewriter both on-line and off-line in conjunction with the paper tape reader and punch.

The Key-board is fitted with a lock that is released only when:

i) The typewriter is off-line, or

ii) The typewriter is on-line, selected for input and the character buffer is not ready ($\overline{\text{CBUSØ}}$)

This lock therefore stops the operator typing further characters to the character buffer until the first has been sent to the C.P.U.

The typewriter has been fitted with push button/indicator unit. The contacts of the push button provide an interrupt (at present L_3). The lamp is illuminated when the typewriter is on-line and selected for input (i.e. the WIS-flip-flop) is set; thus indicating to the operator that he may start to type.

To enable the programs to synchronize the transmission of data with the mechanical operation of the typewriter, two SKS's are provided by the S.C.B.

SKS '2000 - One of the devices (Typewriters, Punch or Reader) on the S.C.B. is Busy i.e. it is not ready to handle a new character. This SKS is normally used for output transfers.

SKS '14000 - The S.C.B. is Ready. This SKS is normally used for input transfers, in which case it means a new character has been put into the buffer by the device.

There are three OCP's involved with operating the typewriter.

OCP '2000 - Select Typewriter for Input, enable the S.C.B., and reset its RDY flip-flop.

OCP '2010 - Select Typewriter for Output, enable the S.C.B. and set its RDY flip-flop.

OCP '2070 - Deselect all devices on the Standard S.C.B. Note this does not change the state of the Enable or RDY flip-flops of the Buffer.

Note 1. Disabling the S.C.B. by enabling another category B channel, does not deselect any device on the Buffer that was already selected.

2. If any of the S.C.B. Devices have been in use, before giving any of the above OCP's, a check must be made as to whether any of the output devices is still operating (i.e. Busy) (by the SKS '2000); or whether there is a character from one of the input devices waiting in the S.C.B. (by the SKS '14000).

6.1.1 Output Routine Example

A simple routine to print out, as an eight octal character number the word in location NUM, would be:-

```
   SKS '2000    Wait for any character
   JMP *-1      devices that may be operating.
   LDX -8,3     Character count to IX3.
   LDB NUM      Output word into B Register.
   OCP '2010    Select Typewriter for O/P, Enable S.C.B.
L1 CRA          Clear A Register
   LLR 3        Shift octal character into A.
   OTA          Output character.
   SKS '2000    Wait for typewriter to
   JMP *-1      finish printing.
   JXI L1,3     Add one to IX3, finished word?
   OTM ='76     Yes, Output "Carriage Return".
   SKS '2000    Wait for typewriter to
   JMP *-1      finish.
   OCP '2070    Deselect Typewriter
```

6.1.2 Input Routine Examples

A simple routine to input eight octal characters from the typewriter to location NUM, would be:-

```
   SKS '2000
   JMP *-1
   CRA          Clear NUM
   STA NUM
   LDX -8,3     Character count to IX3
   OCP '2000    Select Typewriter for I/P, enable Character Buffer
L1 SKS '14000   Test if Buffer Ready
   JMP *+2      Ready
   JMP *-2      Not Ready
   INA          Input Character to a Register
   ANA NUM      "And" partially built up word with new character.
   JXI *+2, 3   Add one to IX3,is word complete?
   JMP L2       Yes
   LGL 3        No, Shift left.
   STA NUM      Store partial word in NUM
   JMP L1       Loop back for next character
L2 STA NUM      Store complete word in NUM
   SKS '14000   Is Buffer Ready?
   JMP *+2      Yes
   JMP *-2      No
   INA          Input carriage return
   OCP '2070    Disconnect typewriter
```

A more elaborate routine, which would use other channels whilst waiting for a character would be:-

```
    SKS '2000
    JMP *-1
    CRA
    STA NUM
    LDX -8,3
    OCP  '2000
```

Routines using
other channels,
but which periodically
does:
SKS '14000
JST INP

```
INP JMP **
    LDA =1
    LDK          Enables Character Buffer only
    INA
    ANA NUM
    JXI *+2,3
    JMP L1
    LGL 3
    STA NUM
    JMP *INP     Return to above routines
 L1 STA NUM
    EXIT         Word complete
```

Note the use of the LDK instruction to enable the Character Buffer without changing the state of the Ready flop.

6.2 Remote Typewriter

This is an identical machine to the Console Typewriter (See Section 6.1) and is used in exactly the same way, with the following differences:

1) It cannot be used off-line.

2) Its OCP's are:

OCP '2001 - Select Remote Typewriter for I/P, enable Standard Character Buffer (SCB) and reset RDY

OCP '2011 - Select Remote Typewriter for O/P, enable SCB and set RDY.

OCP '2071 - Deselect the Remote Typewriter.

3) The interrupt button is connected to level L2.

6.3 Paper Tape Punch

The Facit Paper Tape Punch (8 holes, 150 characters/sec) is connected via the Paper Tape Control Unit to the Standard Character Buffer (SCB). The Punch can be operated on-line or off-line in conjunction with the console typewriter and reader. There are Run-Out controls on the punch itself and the DDP-224 control panel. A Stop Code Character can be punched out by depressing a push button on the control panel.

To enable the programs to synchronize the transmission of data with the mechanical operation of the punch, the following SKS is provided by the S.C.B.

SKS '2000 — One of the devices (Typewriters, Punch or Reader) on the S.C.B. is Busy i.e. it is not ready to handle a new character.

There are four OCP's involved in operating the paper tape punch:

OCP '2200 — Select the Paper Tape Punch, enable the S.C.B., and set its RDY flip-flop.

OCP '2220 — Punch one sprocket hole on the Paper Tape Punch. The state of the S.C.B. is not changed.

OCP '1000 — Punch a Stop Code character on the Paper Tape Punch. The state of the S.C.B. is not changed.

OCP '2070 — Deselect all devices on the S.C.B.
NOTE this does not change the state of theEnable or RDY flip-flops.

Note 1. Disabling the S.C.B. by enabling another category B channel, does not deselect the punch if it was already selected.

2. If any of the S.C.B. devices have been used, before giving any of the above OCP's, a check should be made as to whether

any of the output devices is still operating (i.e. Busy), by SKS '2000 etc.

3. Using OCP '2220 or OCP '1000 deselects the Punch from the S.C.B.

6.3.1 Punch Routine Example

A simple routine that would punch out the contents of location NUM, followed by a Stop Code and run out of about 2" would be:

```
    SKS '2000        Wait for any character device
    JMP *-1          that may be operating
    LDX -8,3         Load IX3 with character count
    OCP '2200        Select Punch and enable S.C.B.
    LDB NUM          "Output" word to B. Reg.
L1  CRA              Clear A
    LLR 3            Shift octal character into A
    OTA              Output Character
    SKS '2000        Wait for punch
    JMP *-1          to complete its cycle
    JXI L1,3         Add one to IX3, is wordfinished?
    OCP '1000        Yes, punch out stop-code
    SKS '2000        Wait for punch
    JMP *-1
    LDX -20,3        Load length of run-out into IX3.
L2  OCP '2220        Punch a sprocket hole
    SKS '2000        Wait for punch
    JMP *-1
    JXI L2,3         Add one to IX3, Run out complete?
    OCP '2070        Yes, deselect Punch.
    EXIT.
```

6.4 Paper Tape Reader

The Facit Paper Tape Reader (it will read 6, 7 or 8 hole tape at up to 1000 characters/sec) is connected via the Paper Tape Control Unit to the Standard Character Buffer (S.C.B.). The Reader can be Operated on-line or off-line in conjunction with the console typewriter and punch.

To enable the programs to synchronize the transmission of data with the mechanical operation of the reader, the following SKS is provided by the S.C.B.

SKS '14000	The S.C.B. is Ready, i.e. a new character has been read into the S.C.B's Buffer.

There are two OCP's involved with operating the Paper Tape Reader:

OCP '2100	Select and Start tape moving through the Paper Tape Reader, enable S.C.B. and reset its RDY flip-flop.
OCP '2070	Deselect all devices on the S.C.B. NOTE this does not change the state of the Enable or RDY flip-flop of the Buffer.

NOTES

1. Once OCP '2100 has been given, paper tape will continue to be driven through the reader at full speed until an OCP '2070 is given (or the stop-code character is read) whether characters are being read or not. Thus if a program cannot process characters coming in from the reader in a time less than roughly 600 μ sec/character, then the program should stop the reader after each character by OCP '27070.
2. The Reader does not detect as a character, blank tape (i.e. sprocket holes only).

3. When reading characters from paper tape, the S.C.B. checks the parity of each character. If there is an error a flip-flop is set which can be tested and reset by SKS '100 or SKS '40100.

6.4.1 Paper Tape Reader Routine

A simple routine entered by a JST, to keep reading characters into Location CHAR, checking for stop code or parity, then returning could be:

Main Program

```
- - - - - - -
- - - - - - -
SKS '2000        Check that no other device on S.C.B.
JMP *-1          is Busy
OCP '2100        Select and start the Reader, enable S.C.B.
- - - - - - -
- - - - - - -
- - - - - - -
- - - - - - -
- - - - - - -
- - - - - - -
SKS '14000       Is there a character in the Buffer?
JMP *+2          Yes.
JMP Continue.    No
JST INR
JMP SC           Stop Code Return
JMP ERR          Parity Error Return
- - - - - - -    Normal Return
- - - - - - -
- - - - - - -
- - - - - - -
- - - - - - -
```

Reader Routine

```
INR **

     STA TEMP       Preserve A

     STX LDX,3      Preserve IX3

     LDA =1         Re-enable S.C.B.   See Note 1.
     LDK

     INA            Input Character

     SKS '1000      Check if Stop Code

     JMP LDX        Yes.   See Note 2

     IRX INR,3      Increment Return Address

     SKS '40100     Check if Parity Error

     JMP LDX        Yes

     IRX INR,3      Ordinary Character

     STA CHAR

LDX  LDX **,3       Restore IX3

     LDA TEMP       Restore A

     JMP* INR       Return

TEMP PZE **
```

Note 1. If the main program is using channels this LDK is needed to re-enable the.S.C.B. without resetting RDY.

2. There is no need to stop the reader as reading the stop code character automatically does that.

A routine that reads character by character, stopping the tape after each character would be:

```
OCP '2100      Select and Start Reader, enable S.C.B.

SKS '14000     Is there a character in the Buffer?

JMP *+2        Yes

JMP *-2        No

INA            Input Character

OCP '2070      Stop Reader
```

6.5 The Orion

The Direct Data Connection (DDC) of the Orion is connected through a special interface unit to DMA #1 of the DDP-224. Transfers occur in both directions between the DDP-224 and the Orion, the 24 bits of the DDP going into or coming from the least significant 24 bits of the 48 bit Orion Word. The transfer rates are:

1 word/32 μsec into Orion

1 word/32 μsec into DDP-224, after a delay of 128 μsec before the first word.

Details of the use of a DMA are given on P.3-6 of the Programmers Manual (71-261A).

The DMA #1 and Interface Unit provide the following SKS's for synchronization.

SKS '26000 - DMA #1 is Busy i.e. it is enabled and therefore has not completed the transfer of a block of words.

SKS '20010 - The Orion DDC is selected for I/P from the DDP-224. Note this does not mean that a transfer is actually in progress.

The following OCP's are used to operate the DMA #1 and Interface Unit:

OCP '6000	Enable DMA #1 and connect its "set-up" channel to O/P Bus.
OCP '1010	Disable DMA #1 and abandon transfer in progress.
OCP '6100	Connect the Range Register of DMA #1 to the I/P Bus.
OCP '1031	Give a "Buffer Overflow Interrupt" to Orion
OCP '1032	Give an "Accept" Interrupt to Orion
OCP '1033	Give a "Disable" Interrupt to Orion
OCP '1034	Give an "End of Frame" Interrupt to Orion
OCP '1030	Clear all the Interrupts to Orion

The Interface Unit provides an Interrupt (on level L_{13}), when the Orion DDC starts on O/P block transfer to the DDP-224.

Note

1. The OCP's '1031 to '1034 set d.c. levels on the associated Orion Interrupt lines, until cleared by an OCP '1030, which must be given before any new interrupt can be sent to the Orion.

2. After giving an End of Frame Interrupt to Orion, it must be cleared, before any new transfers are started between the DDP-224 and the Orion, otherwise it will produce an unintentional Orion interrupt.

3. The Orion DDC "remembers" the identity of all interrupts except the Disable which must be kept on until OMP has examined it.

6.5.1 DDP to Orion Transfer, Program Example

A simple program to transfer '100 words in a block starting at location DATA from the DDP to the Orion would be:

```
     OCP '1010        Abandon any other transfer on DMA ≠ 1.
     OCP '6000        Enable DMA ≠1,
     OTM ORD1         Send starting address to DMA and select O/P Mode
     OTM ORD2         Send word count to DMA
     SKS '26000       Is DMA ≠ 1 still busy?
     JMP *-1          Yes, continue to wait.
     -----            No, transfer is completed.
ORD1 MZE DATA         1st control word for DMA, selects O/P mode and
                      starting address DATA
ORD2 OCT 100          2nd control word for DMA, gives number of words to
                      be transferred.
```

6.5.2 Orion to DDP Transfer, Program Example

A simple program to transfer '100 words from the Orion into a block of DDP storage starting at location DATA would be:

```
     OCP '1010     Abandon any other transfer on DMA #1
     OCP '6000     Enable DMA #1
     OTM ORD1      DMA control words
     OTM ORD2
     SKS '26000    Is DMA #1 still busy?
     JMP *-1       Yes, continue to wait.
     -----         No, transfer is completed.
ORD1 PZE DATA      1st control word for DMA, selects I/P mode and
                   starting address DATA
ORD2 PZE '100      2nd control word for DMA, gives number of words
                   to be transferred.
```

6.5.3 Interrupt to Orion, Program Example

A simple program to give an interrupt to Orion and to clear it again would be:-

```
OCP '103x      Give interrupt, x =1, 2, or 4
LDX -5,3
JXI *,3        Delay for approximately 15 µsec.
OCP '1030      Clear interrupt
```

6.5.4 Interrupt to DDP from Orion, Program Example

In this example, when the Orion sets up a 5 word O/P transfer to the DDP, it gives an interrupt to the DDP, this causes the DDP to enter an interrupt routine that sets up a 5 word I/P transfer on DMA # 1 to a block of store starting at BUF.

```
'47 JMP INT        Interrupt L13 transfer location
INT OCP '6000      Enable DMA #1
    OTM CW1
    OTM =5         5 words
    SKS '26000     Wait for transfer
    JMP *-1        to finish
    JRT '46        Exit from interrupt routine
CW1 PZE BUF        Control word, I/P mode, address BUF
```

6.6 Teletype Teleprinter

The teletype teleprinter, capable of 10 characters/sec is fitted with a reader and a punch (also capable of 10 characters/sec). The teletype transmits data to and from the DDP-224 via the optional character Buffer (O.C.B.) and its Interface and Control Unit. The teletype is situated at about 1,000 feet from the DDP-224. From the programming point of view there is no difference between data coming from the reader or the Key-board. Similarly there is no difference between sending data to the punch or the printer mechanism.

To enable the program to synchronize the transmission of data with the operation of the teletype, the O.C.B. provides the following SKS.

SKS '37041 The Optional Character Buffer is Ready, i.e. either the Buffer has a new character ready to transfer into the CPU, or the buffer is now free to receive a new character from the CPU.

To operate the teletype there are three OCP's:

OCP '1012 Select and start Teleprinter for I/P, enable O.C.B. and reset its RDY flip-flop.

OCP '1013 Select and Start Teleprinter for O/P, enable O.C.B. and set its RDY flip-flop.

OCP '1024 Deselect and Stop theTeleprinter. Note the states of the Enable and RDY flip-flops are not changed.

A push-button is fitted to the Teletype to give an interrupt on level L_1.

Notes

1. Disabling the O.C.B. by enabling another category B channel, does not deselect the Teletype.

2. The Teletype only uses 5 bits/character instead of the normal 6 bits. Therefore on input it is necessary to use a masked INA instruction

   INAM '37

   to ensure that only the 5 bits from the teletype go into the A Register. Obviously the Teletype has a different character code, to the standard DDP-224 code, namely standard Ferranti 5-hole code.

3. Before changing from I/P to O/P mode (or vice versa) it is necessary to deselect the teletype. A delay of approximately 100 msec after the last character has been transmitted, must be allowed before deselecting. A further 100 msec delay must be allowed after deselection before selecting a new mode.

4. There is no parity checking available for the teletype.

The programming for the Teletype is very similar to that for the console typewriter (see section 6.1 etc.). The only differences are:-

a) The use of OCP '1012 for OCP '2000
   OCP '1013 for OCP '2010
   OCP '1024 for OCP '2070
   and SKS '37041 for both SKS '2000 and SKS '14000.

b) The insertion of delays before deselecting and changing mode. For example for changing between I/P mode and O/P mode, the following program could be used:

```
SKS '37041      Is last I/P character in Buffer
JMP *+2         Yes
JMP *-1         No
INAM '37        Input last 5 bit character (see Note 2 above)
LDX 1,3
JXI *,3         Delay for approximately 100 µsec
LDX -6000,3
JXI *,3
```

OCP '1024	Deselect teleprinter and stop motor
LDX 1,3	Delay for approximately 100 msec.
JXI *,3	
LDX -6000,3	
JXI *,3	
OCP '1013	Select for O/P, start motor.
OTM CHAR	Output first 5 bit character
SKS '37041	Has first character been received by the teleprinter?
JMP *+2	Yes
JMP *-2	No

6.7 <u>CRT Film Scanner</u>

The CRT Film Scanner (CRTFS) can be considered as having two sections; the Command Section, which receives 6 bit words (1 word/12 secs) from the Optional Character Buffer (OCB) and a Data Section which transmits 24 bit words into I/P half of DMA #2. (Peak data rate 225 Kc/s).

6.7.1 Command Section

To synchronize the transmission of data from the OCB to the Command Section the following SKS's are provided:

SKS '37041	The O.C.B. is Ready, i.e. the Command Section and the O.C.B. are Ready to receive a new character.
SKS '20000	The film has been advanced the specified number of frames and is now positioned and clamped in the film gate ready for measurements.
SKS '20001	The CRTFS's stage is fully forward
SKS '20002	The CRTFS's stage is fully reversed

For the operation of the Command Section, the following OCP's are provided:

OCP '1014	Select for O/P to the CRTFS, Enable OCB, and set its RDY flip-flop.
OCP '1015	Move film forward by the number of frames given by 6 bit word received from OCB
OCP '1016	Move film back by the number of frames given by 6 bit word received from OCB
OCP '1017	Move film to the next part of the frame
OCP '1020	Move CRTFS stage forward
OCP '1021	Move CRTFS stage backward
OCP '1022	Stop moving the CRTFS stage
OCP '1023	The 6 bit word received from the OCB specified the Scan Line Angle.

Note: OCP's '1015, '1016 and '1023 are given just after a 6 bit word has been sent from the OCB to the Command Section.

6.7.2 <u>Command Section, Program Examples</u>

a) A program to make the CRTFS Film Transport move forward 10 frames could be:

OCP '1014	Select CRTFS and enable OCB
OTM =10	O/P 10 to the CRTFS
OCP '1015	Move the film forward by the specified number of frames
SKS '20000	Has film been advanced the specified number of frames
JMP *+2	Yes
JMP *-2	No, continue to wait

b) A program to make the CRTFS Film Transport move the film to bring the next part of the frame (3 parts per frame) into the measuring position, would be:

OCP '1017	Move film to next part of the frame
SKS '20000	Has film motion ended?
JMP *+2	Yes
JMP *-2	No, continue to wait

c) A program to move the stage forward and measure a picture could be as follows:

OCP '1020	Move CRTFS Stage forward
Program to set up interrupt and start 1st data transfer	
SKS '20001	Is stage fully forward?
JMP *+2	Yes
JMP *-2	No, continue to wait. See note below
OCP '1022	Stop the stage

Note:

Whilst in the SKS loop waiting for the stage to complete its forward traverse, this program will be interrupted at the end of each scan line of the CRTFS, to set up a new transfer.

d) A program to select Scan Line Angle 24, could be:-

OCP '1017	Select CRTFS and enable OCB
OTM =24	O/P 24 to CRTFS
OCP '1023	Select Scan Angle given by the number just transmitted.

6.7.3 Data Section

To synchronize the transmission of data from the CRTFS to the I/P half of DMA #2, the following SKS is provided:

SKS '26001	DMA #2 Busy i.e. the block transfer is not yet complete.

For the operation of the DMA #2, the following OCP's are provided:

OCP '6001	Enable DMA # 2, prepare it to receive control words via the "set-up" channel, connect the "Set-up" channel to the O/P Bus.
OCP '6101	Connect the Range Register of DMA #2 to the I/P Bus.
OCP '1011	Abandon current block transfer - Disable DMA #2.

At the end of each scan line (approximately every 2 msec) the CRTFS generates an interrupt on level L_7.

Note 1. Selecting the DMA #2 for I/P-Mode, automatically selects the CRTFS as the active device on this channel.

2. If more than two control words are transmitted to the DMA via the "set-up" channel, an I/O-hold will result.

6.7.4 Data Section, Program Examples

a) Reading of data from the CRTFS is done by block transfers through DMA #2. This is usually done by an interrupt routine that, is entered at the end of each scan line, which terminates the current transfer and initiates a new block transfer into a new buffer.

```
INT OCP '1011      Abandon previous transfer

    +---------------------+
    | Compute starting    |
    | address of next     |
    | buffer and plant    |
    | in ADD              |
    +---------------------+

    OCP '6001      Enable DMA #2
    OTM ADD        Select I/P mode and starting address
    OTM =50        No. of words greater than number
                   expected on a scan line
    JRT '36        Exit from interrupt routine
ADD PZE **         Control word for input and
                   starting address.
```

b) To read the contents of DMA #2 range register the following program could be used:

```
    OCP '1011      Abandon any current transfer
    OCP '6101      Connect Range Register to I/P-Bus
    INA            Transfer Range Register contents to A
                   Register
```

6.8 Digital CRT Display and Light Pen

The Digital Display CRT (DDCRT) can display a small spot of light (pulsed on for 2 μ sec) anywhere on 1,024 x 1,024 grid on a 21" CRT. The deflection time between any two points on the grid is 12 μsec. The brightness of the spot can be one of four digitally selected levels. The DDCRT is one of the devices which receives 24 bit words from the O/P half of DMA #2. The DDCRT interprets each 24 bit word which it receives as the (X,Y) position on the grid at which the spot is to be displayed and its brightness. The word format is:

1	2 — 11	12 — 13	14 — 23	24
0	X-Coord.	Intensity	Y-Coord.	0

Bit 12	Bit 13	Intensity
0	0	Minimum
0	1	Low
1	0	Medium
1	1	Maximum

To synchronize the transfer of data from the DMA #2 with the operation of the DDCRT, the following SKS is provided:

SKS '26001 DMA #2 is Busy, i.e. a block transfer is in progress.

To operate the DDCRT, the following OCP's are provided:

OCP '6001 Enable DMA #2, connect its "set-up" channel with O/P-Bus

OCP '6101 Connect the Range Register of DMA #2 to I/P-Bus

OCP '1011 Disable DMA #2, disable any current transfer

OCP '1026 Select DDCRT as active device on DMA #2's O/P half.

Note OCP '1026 must be given before the block transfer from DMA #2 is started. Once given it need not be repeated unless another device on the O/P half of DMA # 2 has been selected.

In addition to the DDCRT, there is a Light Pen available. This is a "pen-like" device which when held against the screen can sense the pulse of light from a spot displayed underneath its tip. When a pulse of light is sensed it sets a flip-flop whose state can be examined by an SKS instruction and reset by an OCP.

SKS'20011	Light Pen Flip-Flop is set.
OCP '1027	Reset Light Pen Flip-Flop.

Thus to use the light pen one has:

```
SKS '20011     Has spot been sensed?
JMP *+2        Yes.
JMP NSPT       No, continue search
OCP '1027      Reset Light Pen
```

6.8.1 Digital Display, Program Example

To obtain a display, a block of words describing the required picture in the correct format (See 6.8) has to be assembled, this is then repeatedly transmitted to the DDCRT, for as long as the display is required. To obtain a "flicker-free" display, it is necessary to regenerate it approximately 30 times/sec. With a deflection time of 12 μsec/point, this limits a display to approximately 3,000 points. A routine to keep generating a display given by 3000 words in a block starting at PIC, whilst Sense Switch 1 is Up, would be:

```
   OCP '1026      Select DDCRT
L1 OCP '6001      Enable DMA #2, connect its "set-up" channel to
                  O/P Bus
   OTM CW         Select O/P mode and give starting address
   OTM =3000      3000 words to be transferred
   SKS '26001     Is transfer complete?
   JMP *-1        No, DMA #2 still busy, continue to wait
   SKS 1          Yes, is SW1 up?
   JMP L1         Yes, continue displaying
   JMP EXIT       No, stop displaying
CW MZE PIC        O/P Mode, starting address PIC.
```

6.9 Magnetic Tape System

The DDP-224 has two Datamec Tape Units, capable of writing and reading IBM $\frac{1}{2}$" compatible tape at 25 K char./sec. Full details of these Tape Units (MTU) are given on P.3-21 of the Programmers Manual. These two tape decks share a single Magnetic Tape Control Unit (TCU) and Word Forming Buffer (WFB). Therefore except for rewinding only one tape unit can be selected and operating at a given time. From the point of view of writing to tape, words are transferred to the WFB from the O/P half of DMA #2. From the point of view of reading from tape the WFB transfers words to the CPU via the I/P-Bus.

The standard OCP's and SKS's are listed on P. A-7, A-8, A-9, B-2, B-3 and their use described on P.3-20 to 3-27 of the DDP-224 Programmers Manual (71-261A).

The extra OCP's needed to operate this "non-standard" tape system are:

OCP '6001	Enable DMA #2, connect its "set-up" channel to the O/P-Bus.
OCP '6101	Connect the Range Register of DMA #2 to the I/P-Bus.
OCP '1011	Disable DMA #2, abandon any current transfers.
OCP '1025	Select the Magnetic Tape System for I/P mode.
OCP '54	Connect the Interrupt 10 bit Range Register to the O/P-Bus.

The extra SKS needed to synchronize data transmission with the operation of the MTU, is:

SKS '26001 DMA #2 is Busy i.e. a transfer is still in progress.

With the Tape system selected for I/P, everytime the WFB has a word ready to be transferred into the CPU, it gives an interrupt on level L_{17}. This interrupt is normally a single execute interrupt, however by removing the link from the pin F3A02-06 this interrupt becomes a normal priority interrupt.

Note

1. The OCP's selecting the required WFB mode and Tape Writing operation must be given immediately before the DMA #2 transfer is set up.
2. The OCP '1025 to select I/P mode for the Magnetic System must be given before the OCP's to select WFB mode and Tape Reading operation.
3. If the Range Register is to be used to control reading of tape it must be loaded with the word count before tape motion is started.
4. Giving an OCP that causes the writing of record on tape automatically selects the O/P half of DMA # 2 for operation with the W.F.B.

6.9.1 Magnetic Tape Program Examples

a) An example of a program to write a Binary record of 100 words, four characters per word, with five retries if parity errors occur, could be:

```
WTR CLA
    STA EC      Set parity error counter to zero
WRR OCP '340    Enable WFB for 4 character/word O/P
    OCP '20200  Write BIN Record on MTU #1, start tape moving
    OCP '6001   Enable DMA # 2, connect its "set-up" channel
                to O/P-Bus
    OTM CW      Select O/P mode and starting address
    OTM =100    100 words to be transferred
    SKS '31000  Is the TCU still busy?   See Note 1.
    JMP *-1     Yes, continue to wait
    SKS '70300  No, record has been written, has parity error
                occurred?   See Note 2
    JMP ERR     Yes, jump to error routine
    SKS '71200  No, has the end of the tape been reached?
                See Note 3
    JMP EOT     Yes, jump to end of tape routine
    JMP EXIT    No, record written and checked
ERR OCP '30000  Back space one record
    SKS '31000  Has tape motion ended?
    JMP *-1     No.
    IRX EC      Yes.  Increment error count
    SKN =5      Is error count now five
    JMP BT      Yes, jump to Bad Tape Routine
    JMP WRR     No, try writing record again
 CW MZE DATA    Select O/P mode, starting address DATA
```

Notes

1. The state of the TCU, rather than that of DMA ≠ 2 is sensed, because the DMA transfer will finish before the tape system writes "end of record" and does the longitudinal parity check. So to ensure that the parity check will be valid, the program must wait until the TCU is not Busy.
2. SKS '70300 is used rather than SKS '30300, because this clears the parity error condition as well as sensing it.
3. SKS '71200 is used rather than SKS '31200, because this clears the end of tape condition as well as sensing it.

b) Having tried to write a record without parity error and failed five times, one can try erasing that region of the tape (so that it will be automatically spaced over during reading) and then write the record on a new piece of tape. This erasure can be accomplished by a sequence of 'write end of file - backspace record's to give a suitable length of blank tape. Writing an "end of file" produces about 3.7" of blank tape followed by an "end of file mark" record. The tape is backspaced each time so that the next length of blank tape over-writes the "end of file mark" record giving continuous blank tape. For completeness the length of blank tape should be checked to see if it gives parity errors by back spacing an extra record (i.e. right through the blank tape) after the last 'write end of file - backspace record'. Then if there are no parity errors the tape is brought forward to the end of file mark and backspaced so that the next <u>new</u> record over-writes the "end of file mark". A routine to produce about 25" of blank tape and check this blank tape could be:

```
BT LDX -7,3
   OCP '20400      Write end of file on MTU # 1
   SKS '31000      Has Tape motion finished?
   JMP *-1         No
   OCP '30000      Yes, backspace a record on MTU # 1
   SKS '31000      Has tape motion finished?
   JMP *-1         No
   JXI BT+1,3      Yes, do this seven times
   SKS '70300      Reset parity error
   NOP
   OCP '30000      Backspace a record to test for parity errors in
                   blank tape
   SKS '31000      Has tape motion finished?
   JMP *-1         No
   SKS '70300      Yes, has there been a parity error?
   JMP NEP         Yes, jump to non-erasable tape routine
   OCP '30200      No, skip one record forward
   SKS '31000      Has tape motion finished?
   JMP *-1         No
   OCP '30200      Yes, skip one record forward through blank tape
   SKS '31000      Has tape motion finished?
   JMP *-1         No
   SKS '70300      Yes, has a parity error been found?
   JMP NER         Yes, jump to non-erasable tape routine
   OCP '30000      No, backspace one record (i.e. end of file mark
                   record)
   SKS '31000      Has tape motion finished?
   JMP *-1         No
   JMP WTR         Yes, ready to try writing again (see a) above)
```

c) If in a) above the end of tape mark were sensed, the tape would have to be rewound thus:-

```
EOT OCP ' 30400      Rewind MTU #1
    SKS '31100       Is tape still rewinding?
    JMP *-1          Yes, continue to wait
    HLT '777         No. halt for new tape to be loaded
```

d) An example of a program to read in a 100 words from a BCD record containing more than 100 words, using 4 characters/word and the single execute interrupt, would be:

```
 RT CRA
    STA EC           Zero parity error count
RTR LDA INST
    STA '56          Plant the input instruction in interrupt
                     trap location.   See Note 1.
    OCP '54
    OTM =100         Load the interrupt range register with 100.
                     See Note 2
    OCP '50
    OTM ='77777
    ENB              Enable magnetic tape single execute interrupt
    OCP '1025        Select I/P mode for magnetic tapes
    OCP '341         Enable WFB for 4 character/word I/P mode
    OCP '20100       Read one BCD record from MTU #1 start tape
                     moving
    SKS '31000       Has tape motion finished?
    JMP *-1          No, continue waiting
    SKS '70300       Yes, has parity error occurred?
    JMP RPE          Yes, go to retry routine
    SKS '71200       No, has end of tape been sensed?
    JMP EOT          Yes, go to end of tape routine.
```

```
        JMP EXIT            No, record read satisfactory
    RPE OCP '30000          Backspace one record
        SKS '31000          Has tape motion finished?
        JMP *-1             No
        IRX  EC             Yes, increment error count
        SKN =5              Is error count now five?
        JMP PAR             Yes, tape cannot be read
        JMP RTR             No, try reading again
   IWST INM BUF             Instruction executed by interrupt.  BUF
                            is starting address of block for the words
                            read in.
```

NOTE

1. The address of the INM instruction is incremented each time it is executed, thus the 100 words from the record will be read in off tape into BUF to BUF +99.

2. As each word is read in off tape the range register is decremented by one until it effectively reaches one, when it inhibits further interrupts from the TCU.
   Thus the first 100 words will be read in and the remaining words in the record will be spaced over and ignored. The tape will stop in the end of record gap.

e) An example of a program to read in a Binary record from tape, of some unknown number of words less than 1,000. The characters are to be packed at 4 per word and a normal priority interrupt routine is to used.

```
        CRA
        STA WC              Zero Word count
        TAX 0,3             Zero IX3
        LDA INS
        STA '57             Plant jump to interrupt routine
```

```
         OCP '54           Load interrupt range register
         OTM =1000         with 1,000
         OCP '50
         OTM ='77777
         ENB               Enable magnetic tape interrupt
         OCP '1025         Select I/P mode for magnetic tapes
         OCP '341          Enable WFB for 4 character/word I/P mode
         OCP '20000        Read on Binary record from MTU #1, start
                           tape moving
         SKS '31000        Has tape motion finished?
         JMP *-1           No, continue waiting
         JMP EXIT          Yes, record has been read
     INS JMP INR           Instruction planted at interrupt transfer
                           location
```

<u>Interrupt Routine</u>

```
     INR OCP '1025         Select I/P Mode for magnetic tapes.   See
                           Note 2.
         INM BUF,3
         IRX WC,3          Increment word count and IX3, See Note 3.
         JRT '56           Exit from interrupt routine
```

<u>Note</u>

1. For brevity parity checking etc. has been left out of this example.
2. If this OCP '1025 is used at the beginning of the interrupt routine, then it is possible whilst waiting for the record to be read, to carry out other transfers involving the I/P or O/P-Buses.
3. When the record has been read WC will contain the number of words in the record.

6.10 Storage Tube Display (STD)

This device is means of producing simple displays (e.g. histograms) on a Standard Tektronix 5" Storage Tube Oscilloscope (Type 564). Points can be displayed on the screen of the CRT on a grid 256 x 256. The display rate is approximately one point/10 μ sec. The display cannot be cleared from the screen by program. The display unit receives data from the DDP-224 via the parallel O/P channel. The display is situated at about 1200 feet from the DDP-224.

The words sent to the display define the required (X,Y) position of points, one point/word. The format of the word is:

1 8	9 16	17 24
Not used	X-Coord.	Y-Coord.

To simplify attaching this display to the DDP-224, it was decided not to make use of a control back from the display (DEVLD pulse) every time it received a word. Instead whilst the STD is selected on the Parallel O/P channel the RDY flip-flop is held set. Therefore the synchronizing of the data transfers cannot be done by SKS '11000, sensing whether the channel is ready. Instead a time delay of approximately 10 μsec has to be arranged in between each transfer.

To operate the S.T.D. the following OCP is provided

```
   OCP '2          Select Storage Tube Display and Enable the
                   Parallel O/P Channel
```

Thus a program to produce a display of 100 points, described by the words in a block starting at PIC, would be:

```
   LDX -100,1      Load IX1 with 100.
   OCP '2          Select STD and Enable Parallel O/P Channel
L1 OTM PIC+100,1   Output a point
   LDX -3,3
   JXI *,3         Wait roughly 10 µsec
   JXI L1,1        Repeat  for 100 points
```

6.11 The Hough Powell Device (HPD)

The HPD can be thought of in two sections - the Command Section and the Data Section. The command section (roughly 4 word/15 secs) receives 18 bit words from the DDP-224 through the Parallel O/P Channel (which it shares with the STD. See 6.10). The Data Section transmits 18 bit words to the DDP-224 through the Parallel I/P Channel #1 (roughly 4,000 word/sec. in 10 second bursts).

6.11.1 HPD Command Section

This is the section of the HPD which receives and interprets 18 bit "commands" received from the DDP-224. These commands control the film transport, stage motion and digitizing systems within the HPD. Each time the HPD has completed the measurements on a given picture a sequence of four commands is transmitted to the HPD to:

reset the stages
advance the film
select the scan mode, stage speed and scan line density for the next measuring scan of a picture.

To synchronize the transmission of words to the HPD with the operation of its command section the following SKS is provided:

SKS '11000 Parallel O/P Channel is ready to transmit another word to the HPD

NOTE In general the Command Section can receive a command at any time (except for 16 μsec after the receipt of command) and, in general, a command over-rides earlier commands stopping measurement if it is already in progress. So the issuing of commands has to be synchronized to the HPD operation as a whole by either waiting until a complete frame has been scanned or until the whole of the picture upto given stage coordinate has been received by the computer.

To operate the Command Section of the HPD the following OCP is provided:

OCP '00006 Select HPD, enable Parallel O/P Channel and set its RDY flip-flops.

The format of the command word (*) is:

Note* - see HPD-65-1A

1 — 6	7	8	9	10 11 12	13 — 24
X——X	1	0	Tag	Code	Argument

Code

'0 COORD

'1 Scan in 90^{o}-Mode

'2 End of Film

'3 Move to W

'4 Spare

'5 Scan in 0^{o}-Mode

'6 Move film ± N frames i.e. pictures

'7 Move to, X

Argument

This part of the command is interpreted in various ways (e.g. X-Coordinate, ±Nframes, speed view, density) depending on the command.

6.11.2 Command Section, Program Examples

A routine that could be used to output a sequence of commands to the HPD is for example:

```
    LDX -4,3        -4 to IX3
    OCP '6          Select HPD and enable the parallel O/P channel
    DMB CMD+4,3     Output for command words
    JMP EXIT
CMD OCT 472000      Move to X = '2000
    OCT 433000      Move to W = '3000
    OCT 464020      Move film forward '20 frames
    OCT 453000      Scan in 0° mode, speed 1, view 2, density 0
```

6.11.3 <u>HPD Data Section</u>

The HPD Data Section has a 128 word buffer store, where the data is assembled, before transmission to the DDP-224. At present the programs are arranged to periodically sense whether this buffer is more than half full, at which time 50 words are read as rapidly as possible (approximately 1 word/7μsec) into the DDP-224.

To enable the programs to synchronize the data transfers with the operation of the HPD Data Section, the following SKS's are provided:

SKS '12000	Parallel I/P Channel #1 is Ready i.e. a new word is ready in the channel to be transferred into the CPU
SKS '20003	HPD is Disabled
SKS '20004	HPD is Enabled
SKS '20005	HPD is in the Accept state
SKS '20006	HPD has reached the End of Frame
SKS '20007	HPD Buffer store has overflowed.
SKS '20017	HPD Buffer Store contains more than 64 words.

To operate the HPD Data Section the following OCP's are required:

OCP '00001	Enable Parallel I/P Channel #1 for HPD Data, and reset the RDY flip-flop.

<u>NOTE</u>

The HPD interface is designed so that it always has a data word ready on the input lines to the channel i.e. as soon as the HPD starts to generate digitizings, it sets up a word for the I/P channel. So that the RDY flip-flop of the channel is always set before the channel is enabled for a transfer. If the above OCP were used the RDY flip-flop would be reset, thus loosing a word. Therefore instead of the OCP, the following should be used.

LDA ='10000	Set bit 12 in the A register
LDK	Enable the Parallel I/P Channel # 1, do not change the state of the RDY flip-flop. (See Table 4).

The HPD gives an interrupt on level L_{12}, whenever it sets any of the SKS's '20003 to '20007.

6.11.4 Data Section, Program Example

a) An example of a program which periodically senses to see if there are more than 64 words in the buffer and when there is, reads in 50 words into a block of storage starting at BUF:

```
+------------------+
| Main program     |
| periodically     |
| cycles through   |
| SKS '20017       |   Sense number of words in buffer
| JST RD           |   More than 64, jump to read data
| - - - - - - - -  |   Less than 64 words, continue program
| - - - - - - - -  |
+------------------+

RD JMP **
   LDA ='10000
   LDK               Enable Parallel I/P Channel #1
   LDX -50,3         Load IX3 with -50
   FMB BUF+50,3      Input 50 words into BUF
   JMP *RD           50 words read in return to main
                     program.
```

b) An example of a program to decide the reason for the HPD giving an interrupt:

```
'45 JMP HINT         Jump to interrupt routine, in transfer
                     location
   HINT SKS '20003   Had HPD been Disabled?
        JMP DNB      Yes, exit to associated routine
        SKS '20004   No, has HPD been Enabled?
        JMP ENB      Yes, exit to associated routine
        SKS '20005   No, had HPD Accepted?
        JMP ACP      Yes, exit to associated routine
```

```
SKS '20006   No, has HPD reached End of Frame?
JMP EOF      Yes, exit to associated routine
------       No, start Buffer Overflow routine.
```

6.12 Sonic Spark Chamber System (OLX)

A parallel high speed data system has been installed between the Nimrod Experimental area and the Parallel I/P Channel # 2 of the DDP-224. At present there is only one spark chamber system attached to this system, that of the P3 experiment. Another spark chamber system (OLX # 2) involving a PDP8 computer is to share this transmission system in the near future, but the details have still to be decided.

The present system transmits 16 bit words, in blocks of 128 words and there are approximately 3 blocks/2 secs.

To enable the programs to synchronize the data transmissions with the operation of the Sonic Chamber System, the following SKS is provided:

SKS '37003 Parallel I/P Channel # 2 is Ready i.e. there is a word ready in the channel to be transferred into the CPU.

To operate the OLX # 1 system the following OCP is provided

OCP '00003 Enable Parallel I/P Channel # 2, with sonic spark chamber system # 1 selected. Reset RDY flip-flop.

The OLX #1 generates an interrupt on Level L_4, whenever it has a block of 128 words ready for transmission to the DDP-224.

NOTES

1. After receipt of the interrupt a delay of 40 μsec must be allowed before the OCP to enable the channel is given.
2. After the enabling OCP has been given, it is necessary to allow a delay of 40 μsec before sensing whether the channel is ready (SKS '38003) and then reading a word in. This delay allows the program to detect the end of the block of words which may be shorter or longer than the nominal 128 words. If a new word is not ready in the channel after this 40 μsec delay when it can be assumed that the block is ended. This delay must be allowed between the enabling OCP and the input of the first word.

6.12.1 OLX # 1, Program Example

On receipt of the OLX # 1 interrupt a program is entered which reads in words until the end of the data block (as defined above). The data is read into a block of storage beginning at BUF.

```
'31 JMP INT        Jump instruction in interrupt transfer location
INT LDX -15,3      -15 to IX3
    JXI *,3        Delay for 40 µsec
    OCP '3         Enable Parallel I/P Channel #2 and OLX #1
    LDX -128,2     -128 to IX2
 LI LDX -15,3      -15 to IX3
    JXI *,3        Delay for 40 µsec
    SKS '37003     Is the channel RDY?
    JMP *+2        Yes, jump to read in word
    JMP EODB       No, end of data block, jump to appropriate routine
    INM BUF+128,2    Input word
    JXI L1,2       Increment IX2 and continue
    JMP ERR        Error condition, more than 128 words in a block.
```

6.12.2 OLX #2

As stated above the details of this part of the system has to be decided. It will be sharing the high speed data system with OLX #1. An extra OCP will be supplied to select to OLX #2:

OCP '1035 Enable Parallel I/P Channel #2, select OLX to use the data transmission system. Reset the RDY flip-flop.

An interrupt will be provided for OLX #2, on level L_{06}, for use when it has a data block ready for transfer.

In addition OLX # 2 is to be provided with a remote typewriter similar to that of OLX #1.

6.13 Calcomp Graph Plotter

The Calcomp plotter is a digital incremental plotter. The pen is moved in .01" steps (maximum speed is 300 steps/second). The pen may be also raised and lowered from the paper (maximum of 10 times/sec). The graphs are drawn on a 12" wide continuous strip of paper, which passes over a drum that provides the X displacement. The pen is on a carriage which slides on bars above the surface of the drum, providing the Y displacement.

The Calcomp plotter is controlled entirely by OCP's. No channels are used or disturbed. The OCP's cause displacements of one unit step in the directions shown below, raise and lower the pen.



Pen Up: OCP '2640
Pen Down: OCP '2620

To synchronize the OCP's with the operation of the plotter, the following SKS is provided.

SKS '22600 The Calcomp Plotter is Busy with an operation.

To plot a graph etc., with this device, therefore consists of interpreting it in terms of line elements in eight possible directions and isolated points. This information is then transmitted to the plotter as a string of OCP's in the following manner.

```
SKS '22600      Is the Plotter still Busy?
JMP *-1         Yes, continue to wait
OCP **          No, give OCP
```

7.0 Conclusion

This system has been operating for some time with all the devices described above except for ÒLX #2. The system is operated by a time sharing control program called Pentagon. Pentagon is primarily responsible for operating the system with the typewriters, H.P.D., CRT Film Scanner and remote teleprinter in various combinations, but not all together due to core store space limitations. The magnetic tapes, plotter, CRT Displays and Paper Tape Equipment are operated as auxilaries to the system.

Acknowledgements

Very many people have contributed to the successful setting up of this system. The author would like to thank the following people for their help in programming, hardware installation and fault finding etc.:

Applied Physics Division

Theoretical Group

Mr. P. A. Denny	Mr. J. Sparrow	Mrs. S. Norris
Dr. R. Taylor	Mr. C. J. Collie	Mrs. R. E. Brody
Mr. J. W. Gardner	Mr. A. J. Redman[2]	Mr. M. J. Mitchell
Mr. M.C. Turnill[1]	Mr. J. W. Burren	Dr. A. Accensi

Track Analysis Group

Mr. R. A. Lawes — Mr. J. F. MacEwan — Mr. J. A. Bourne

High Energy Physics Division

P3 Group

Dr. C. W. Whitehead[3] — Mr. L. Bird[3] — Mr. C. Stahlbrandt[4]

K6 Group

Mr. R. MacKenzie — Mr. R. J. Tapper[5] — Mr. H. Shaylor[6]

Electronics Group

Mr. P. E. Smith

K_2^0 Group

Mr. P. Sharp[3] — Mr. H. Norton

Computer Control Co.

Mr. R. McKenzie — Mr. S. Pearson

I.C.T. Ltd.

Mr. P. Price and his team of Orion engineers.

[1] Imperial College, London
[2] Woolwich Polytechnic
[3] A.E.R.E., Harwell
[4] Now returned to CERN, Geneva
[5] Cambridge University
[6] Birmingham University

T A B L E 1

OCP Assignments

Associated Devices	OCP	Description
Typewriters and Paper Tape Equipment	01000	Punch Stop Code On Paper tape punch
	02000	Console Typewriter I/P select and character buffer enable.
	02001	Remote typewriter I/P select and character buffer enable.
	02010	Console typewriter O/P select and character buffer enable
	02011	Remote typewriter O/P select and character buffer enable.
	02070	Disconnect all I/O devices on the character buffer.
	02071	Disconnect remote typewriter.
	02100	Enable paper tape reader and character buffer.
	02200	Enable paper tape punch and character buffer.
	02220	Feed one sprocket hole on paper-tape punch.
Orion	06000	Enable DMA #1
	06100	Connect DMA #1's range register to I/P bus.
	01010	Disable DMA #1
	01030	Clear Interrupts to Orion.
	01031	Send Buffer overflow interrupt to Orion.
	01032	Send Accept interrupt to Orion.
	01033	Send Disable interrupt to Orion.
	01034	Send End of Frame interrupt to Orion.
Remote Teletype Teleprinter	01012	Select teletype for I/P and enable optional char. buffer.
	01013	Select teletype for O/P and enable optional char. buffer.
	01024	Deselect teletype and turn its motor off.
CRT Film Scanner	01014	Select CRT to receive commands and enable optional character buffer.
	01015	CRT Film Transport Forward.
	01016	CRT Film Transport Reverse.
	01017	CRT Film Transport Increment.
	01020	CRT Stage Forward.
	01021	CRT Stage Reverse.
	01022	CRT Stage Stop.
	01023	CRT Select Scan Line Angle.
	06001	Enable DMA # 2.
	06101	Connect DMA #2's Range Register to I/P Bus.
	01011	Disable DMA # 2.
Digital CRT Display	01026	Select Digital CRT Display
	06001	Enable DMA # 2.
	06101	Connect DMA # 2's Range Register to I/P Bus.
	01011	Disable DMA #2.
	01027	Clear Light Pen.

T A B L E 1 (contd.)

Associated Devices	OCP	DESCRIPTION
Magnetic Tape System	01025	Select Magnetic Tapes for I/P.
	06001	Enable DMA #2.
	06101	Connect DMA #2's Range Register to I/P Bus.
	01011	Disable DMA #2.
	00310	Enable WFB for 1 character/word O/P.
	00311	Enable WFB for 1 character/word I/P.
	00320	Enable WFB for 2 character/word O/P.
	00321	Enable WFB for 2 character/word I/P.
	00330	Enable WFB for 3 character/word O/P.
	00331	Enable WFB for 3 character/word I/P.
	00340	Enable WFB for 4 character/word O/P.
	00341	Enable WFB for 4 character/word I/P.
	20000	Read Odd Parity on MTU #1.
	20001	Read Odd Parity on MTU #2.
	20100	Read Even Parity on MTU # 1.
	20101	Read Even Parity on MTU # 2.
	20200	Write Odd Parity on MTU #1.
	20201	Write Odd Parity on MTU #2.
	20300	Write Even Parity on MTU #1.
	20301	Write Even Parity on MTU #2.
	20400	Write End of File on MTU#1.
	20401	Write End of File on MTU # 2.
	30000	Backspace one record on MTU #1.
	30001	Backspace one record on MTU #2.
	30100	Backspace one File on MTU #1.
	30101	Backspace one File on MTU #2.
	30200	Skip one Record on MTU #1.
	30201	Skip one Record on MTU #2.
	30300	Skip one File on MTU #1.
	30301	Skip one File on MTU #2.
	30400	Rewind MTU #1.
	30401	Rewind MTU #2.
H.P.D., Sonic Spark Chambers (OLX) and Storage Tube Display	00001	Enable Parallel I/P Channel #1 for HPD Data.
	00003	Enable Parallel I/P Channel #2 for Sonic Spark(OLX #1)
	01035	Enable Parallel I/P Channel #2 for Sonic Spark Chamber #2 (OLX #2).
	00002	Select Storage Tube CRT Display and Enable Parallel O/P Channel.
	00006	Select HPD to receive commands and Enable Parallel O/P Channel.
Calcomp Graph Plotter	02601	Step-Y or carriage Right.
	02602	Step +Y or carriage Left.
	02604	Step-X or drum Up.
	02605	Step -X and -Y.
	02606	Step -X and +Y.
	02610	Step +X or drum Down.
	02611	Step +X and -Y.
	02612	Step +X and +Y.
	02620	Plotter Pen Down.
	02640	Plotter Pen Up.

T A B L E 1 (contd.)

Associated Devices	OCP	Description
Interrupt System	00050 00054	Connect Interrupt Mask Register to O/P Bus. Connect Interrupt Range Register to O/P Bus.
	01036 01037	Spare Spare

TABLE 2

SKS Assignments

Associated Devices	SKS	Description
Typewriters and Paper Tape Equipment	00100 01000 02000 14000	Parity Error in Standard I/O Character Buffer. Stop Code. Character Device on Standard I/O Character Buffer Busy Standard I/O Character Buffer Ready.
Orion	26000 20010	DMA #1 Busy. Orion DDC Selected for I/P.
Teletype	37041 37141	Optional Character Buffer Ready. Parity Error in Optional Character Buffer.
CRT Film Scanner	37041 37141 20000 20001 20002	Optional Character Buffer Ready. Parity Error in Optional Character Buffer. CRT Film Arrived. CRT Stage Fully Forward. CRT Stage Fully Reversed.
Digital CRT Display	26001 20011	DMA # 2 Busy. Light Pen Set.
Magnetic Tape System	26001 30300 31000 31100 31101 31200 31300 31400	DMA #2 Busy. Parity Error in WFB. Magnetic Tape Control Busy. MTU #1 is Rewinding. MTU #2 is Rewinding. End of Tape Sensed. Beginning of Tape sensed. End of File sensed.
H.P.D. Sonic Spark Chambers (OLX) and Storage Tube Display	11000 12000 37003 20003 20004 20005 20006 20007 20017	Parallel O/P Channel Ready. Parallel I/P Channel #1 Ready. Parallel I/P Channel #2 Ready. HPD Disabled. HPD Enabled. HPD Accept. HPD End of Frame. HPD Buffer Overflow. HPD Buffer holds more than 64 words.
Calcomp Graph Plotter	22600	Graph Plotter is Busy.
	20012 20013 20014 20015 20016	Spare Spare Spare Spare Spare

T A B L E 3

Interrupt Assignments

External Name	Interrupt Level	Trap Location	Transfer Location	Description
	LØØ	20_8	21_8	
ELG	LØ1	22	23	Remote Teletype Teleprinter - Interrupt Button.
ELB	LØ2	24	25	Remote Typewriter - Interrupt Button
ELA	LØ3	26	27	Console Typewriter - Interrupt Button
ELE	LØ4	30	31	Sonic Spark Chamber, OLX # 1.
	LØ5	32	33	
ELD	LØ6	34	35	Sonic Spark Chamber, OLX # 2.
ELC	LØ7	36	37	CRT Film Scanner - End of Scan Line
	L1Ø	40	41	
	L11	42	43	
ELF	L12	44	45	HPD Interrupt.
ELH	L13	46	47	Orion Interrupt.
	*L14	50	51	
	*L15	52	53	
	*L16	54	55	
L17	*L17	56	57	Magnetic Tape System has I/P Word Ready.

Note: * - Single Execute Interrupt

T A B L E 4

Channel Status (LDK/STK) System

Name	Bit Position	Description
E1	1Ø	Optional Character Buffer, Enable.
E2	11	Parallel O/P Channel, Enable.
E3	12	Parallel I/P Channel #1, Enable.
E4	13	Parallel I/P Channel #2, Enable.
E5	14	Magnetic Tape system selected for I/P.
E6	15	DMA # 1, set-up Enable.
E7	16	DMA # 2, set-up Enable.
E8	17	Interrupt Range Register, set-up Enable.
	18	
	19	
	20	
	21	
	22	
	23	
E9	24	Standard Character Buffer, Enable.

TABLE 5

Core Store Layout

Address	Core Module	True Core Location
00000 to 07777	A	00000 to 07777
10000 to 17777	B	10000 to 17777
20000 to 27777	C	20000 to 27777
30000 to 37777	C	20000 to 27777
40000 to 47777	A	00000 to 07777
50000 to 57777	B	10000 to 17777
60000 to 67777	C	20000 to 27777
70000 to 77777	C	20000 to 27777

Note: Modules A and B are a single 8K module, C is a single 4K module



NOTES

1. $\xrightarrow{24}$ - REPRESENTS A GROUP OF DATA LINES, THE NUMBER INDICATES THE NUMBER OF BITS.
2. (N) - INDICATES THAT THE DATA LINES ARE N FEET IN LENGTH.

Fig. 1 DATA FLOW BETWEEN THE CENTRAL PROCESSOR CHANNELS AND DEVICES



NOTES.
1. $\xrightarrow{24}$ -REPRESENTS A GROUP OF DATA LINES, THE NUMBER INDICATES THE NUMBER OF BITS.
2. (N)-INDICATES THAT THE DATA LINES ARE N FEET IN LENGTH.
3. OCP'S 1015, 1016, 1017, 1020, 1021, 1022, 1023.
4. OCP's 310, 311, 320, 321, 330, 331, 340, 341.
5. OCP's 2000, 20001, 20100, 20101, 20200, 20201, 20300, 20301, 20400, 20401, 30000, 30001, 30100, 30101, 30200, 30201, 30300, 30301, 30400, 30401.
6. OCP's 2601, 2602, 2604, 2605, 2606, 2610, 2611, 2612, 2620, 2640.

Fig. 2 DISTRIBUTION OF OCP'S



Fig. 3 DISTRIBUTION OF SKS'S



NOTES
1 24⇒ - REPRESENTS A GROUP OF DATA LINES, THE NUMBER INDICATES THE NUMBER OF BITS.
2 (N) - INDICATES THAT THE DATA LINES ARE N FEET IN LENGTH.
3 THE ELA TO ELH AND L17 ARE INTERRUPTS, FOR LEVEL ASSIGNMENT SEE TABLE 3.
4 THE E1 TO E9 ARE LINES IN THE CHANNEL STATUS SYSTEM, FOR BIT ASSIGNMENT SEE TABLE 4.

Fig. 4 DISTRIBUTION OF INTERRUPTS AND CHANNEL STATUS SYSTEM

available from
Her Majesty's Stationery Office
PO Box 569 London SE1
49 High Holborn London WC1
423 Oxford Street London W1
13a Castle Street Edinburgh 2
109 St Mary Street Cardiff
Brazennose Street Manchester 2
50 Fairfax Street Bristol 1
35 Smallbrook Ringway Birmingham 5
7–11 Linenhall Street Belfast 2
or through any bookseller

printed in England

SO Code No 91-40-0-0